

Fragmentation of habitat affects communities and movement of nekton in salt marsh tidal creeks

Paul J. Rudershausen*, Jeffery H. Merrell, Jeffrey A. Buckel

*Corresponding author: pjruders@ncsu.edu

Marine Ecology Progress Series 586: 57–72 (2018)

Supplement. Code modified from Kéry (2010) for fitting random-intercept binomially distributed generalized linear mixed models (GLMMs) (logit links) via Bayesian inference to data on movement of adult *Fundulus heteroclitus* past road and reference crossings in North Carolina tidal creeks. GLMMs were run through *R* software by calling *JAGS* software. Similar code was used for additional models fitted to other combinations of covariates (see Table 3).

```
model {  
  
##### Priors  
for (i in 1:n.creeks) {  
  alpha[i] ~ dnorm(mu.int, tau.int) #prior for population random intercepts  
}  
  
mu.int ~ dnorm(0,0.000001) #hyperprior for grand mean random intercepts  
sigma.int ~ dunif(0, 100) # hyperprior for standard deviation of random intercepts  
tau.int <- 1 / (sigma.int*sigma.int) #calculated precision for random intercepts  
  
beta1 ~ dnorm(0,0.000001)  
beta2 ~ dnorm(0,0.000001)  
beta3 ~ dnorm(0,0.000001)  
beta4 ~ dnorm(0,0.000001)  
  
##### likelihood  
for (i in 1:n.obs) {  
  C[i] ~ dbin(p[i], N[i])  
  logit(p[i]) <- alpha[CreekCode[i]] + beta1*PercentMarsh[i] + beta2*CurrentVelocity[i]  
    + beta3*CurrentVelocitySquared[i]  
    + beta4*PercentMarsh[i]*CurrentVelocitySquared[i]  
  
##### Posterior predictive check using a chi-square discrepancy measure  
ExpC[i] <- N[i] * p[i]  
E[i] <- pow((C[i]-ExpC[i]),2) / (ExpC[i] + 0.5)  
C.new[i]~dbin(p[i], N[i]) #Create replicate data set  
E.new[i] <- pow((C.new[i]-ExpC[i]),2) / (ExpC[i] + 0.5)  
  
}  
  
fit <- sum(E[]) #sum of squared residuals for actual data set  
fit.new <- sum(E.new[]) #sum of squared residuals for new data set  
test <- step(fit.new-fit) #Bayesian probability value  
  
} #end model
```